version 18.0               // version control
set processors 8           // to ensure replicability across different numbers of cores
clear all                  // clear existing data
macro drop _all            // and macros, clean slate
set seed 20220909          // set seed
set scheme cleanplots      // clean plots
/* 
Notes: 
1) Must put "set sortseed #" right on top of EACH "telasso" (rather than at the top of the do file) to ensure the same number of selected controls will be reported.
2) For detailed explanations, see Stata's technical support's email chain on 10/10/2023. 
*/

local pgm  "dst-Figure_7c_sleep_deprivation_telasso_alt_spec"         // file name
local who  "Muzhe Yang"                                               // author
local dte  "2025-01-20"                                               // created date
local dte2 "`c(current_date)'"                                        // last run date
local tag  "`pgm'.do, created by `who' on `dte', last run on `dte2'"

capture log close
log using "code\analysis\figures\\`pgm'.txt", text replace 
display "`tag'"

**# data prep ------------------------------------------------------------------------------------

use "data_clean\dst-data04_for_estimation_within_250_miles", clear
keep if wave == 2

*## (1) create the 9 cells, following page 215 of the following paper:
*## Giuntella, O. and F. Mazzonna (2019). "Sunset Time and the Economic Effects of Social Jetlag: Evidence from US Time Zone Borders." Journal of Health Economics 65: 210-226.

assert !missing(centroid_lat)
assert !missing(region)
gen     cell = 1 if centroid_lat > 40                        & region == "eastern and central"
replace cell = 2 if (34 < centroid_lat & centroid_lat <= 40) & region == "eastern and central"
replace cell = 3 if centroid_lat <= 34                       & region == "eastern and central"
replace cell = 4 if centroid_lat > 40                        & region == "central and mountain"
replace cell = 5 if (34 < centroid_lat & centroid_lat <= 40) & region == "central and mountain"
replace cell = 6 if centroid_lat <= 34                       & region == "central and mountain"
replace cell = 7 if centroid_lat > 40                        & region == "mountain and pacific"
replace cell = 8 if (34 < centroid_lat & centroid_lat <= 40) & region == "mountain and pacific"
replace cell = 9 if centroid_lat <= 34                       & region == "mountain and pacific"

*## (2) control variables

global x "pop white_pct black_pct hispanic_pct pop_under_18yr_pct pop_65yr_over_pct age_median educ_hs_pct educ_coll_pct married_pct hh_size hh_income_median home_value_median no_health_ins_pct unemploy_pct"
global cvars    c.($x dist_to_border centroid_lat daylight_dur_max)##c.($x dist_to_border centroid_lat daylight_dur_max)
global controls $cvars i.cell i.cell#c.($x dist_to_border centroid_lat daylight_dur_max) 

**# estimation prep -------------------------------------------------

encode CountyFIPS, gen(county_fips)
quietly tab cell, gen(cell_)
egen nomiss = rowmiss($x dist_to_border centroid_lat daylight_dur_max)
assert !missing(StateAbbr)
local radius = 50
gen sample_selection = (nomiss == 0 & StateAbbr != "AZ" & dist_to_border <= `radius')

**# figure ----------------------------------------------------------------------------------------------

set sortseed 12102023
telasso (sleep_depr $controls) (treat $controls) if sample_selection == 1, vce(cluster county_fips) 
estimates store case_1

set sortseed 12102023
telasso (sleep_depr $controls, lasso(selection(bic))) (treat $controls, lasso(selection(bic))) if sample_selection == 1, vce(cluster county_fips) 
estimates store case_2

set sortseed 12102023
telasso (sleep_depr $controls, lasso(selection(cv))) (treat $controls, lasso(selection(bic))) if sample_selection == 1, vce(cluster county_fips) rseed(10101) 
estimates store case_3

set sortseed 12102023
telasso (sleep_depr $controls, lasso(selection(adaptive))) (treat $controls, lasso(selection(bic))) if sample_selection == 1, vce(cluster county_fips) rseed(10101) 
estimates store case_4

set sortseed 12102023
telasso (sleep_depr $controls, ainclude($x dist_to_border centroid_lat daylight_dur_max cell_1-cell_9)) (treat $controls, ainclude(dist_to_border centroid_lat daylight_dur_max)) if sample_selection == 1, vce(cluster county_fips)
estimates store case_5

coefplot (case_1, label("Case 1: Method for determining the value of the LASSO penalty parameter: plugin")) ///
         (case_2, label("Case 2: Method for determining the value of the LASSO penalty parameter: BIC")) ///
         (case_3, label("Case 3: Method for determining the value of the LASSO penalty parameter: CV")) ///
         (case_4, label("Case 4: Method for determining the value of the LASSO penalty parameter: adaptive CV")) ///
         (case_5, label("Case 5: Case 1 plus some predictors being required to be always included in the LASSO (more explanations given in the notes section)")) ///
         , keep(r1vs0.treat) ///
         coeflabels(r1vs0.treat = `""Treat (1/0):" "1 for census tracts" "located east of" "the time zone border;" "0 for census tracts" "located west of" "the time zone border""', labsize(*0.8)) ///
         mlabels mlabsize(*0.8) mlabcolor(black) mlabformat(%9.3f) mlabposition(12) mlabgap(*2) ///
         xline(0, lpattern(dash) lwidth(*2) lcolor(gray)) ///
         levels(95 90) ciopts(lwidth(*3 *3) lcolor(*.3 *.6)) ///
         xlabel(-0.5(0.5)3.5) ///
         ylabel(, noticks) ///
         legend(size(*0.75) position(6)) ///
         title("Dependent variable: sleep deprivation", size(*0.8) span) ///
         subtitle("The dependent variable is defined as the prevalence of adults aged 18 years or older who report usually getting insufficient sleep (i.e., less than 7 hours, on average, during a 24-hour period)." ///
                  "The prevalence is at the census tract level, measured annually, and measured in 0–100 percentage points.", size(*0.6) span) ///
         note("Point estimates and the confidence intervals at the 95% level and the 90% level (in darker color) are reported in the figure.", size(*0.8)) ///
         scale(0.7)

graph save "code\analysis\figures\\`pgm'", replace
graph export "code\analysis\figures\\`pgm'.png", replace
graph export "code\analysis\figures\\`pgm'.emf", replace

log close
exit